

C06GBF – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

C06GBF forms the complex conjugate of a Hermitian sequence of n data values.

2 Specification

```
SUBROUTINE C06GBF(X, N, IFAIL)
  INTEGER          N, IFAIL
  real           X(N)
```

3 Description

This is a utility routine for use in conjunction with C06EAF, C06EBF, C06FAF or C06FBF to calculate inverse discrete Fourier transforms (see the Chapter Introduction).

4 References

None.

5 Parameters

- 1:** X(N) — *real* array *Input/Output*
On entry: if the data values z_j are written as $x_j + iy_j$ and if X is declared with bounds (0:N-1) in the (sub)program from which C06GBF is called, then for $0 \leq j \leq n/2$, X(j) must contain $x_j (= x_{n-j})$, while for $n/2 < j \leq n-1$, X(j) must contain $-y_j (= y_{n-j})$. In other words, X must contain the Hermitian sequence in Hermitian form. (See also Section 2.1.2 of the Chapter Introduction).
On exit: the imaginary parts y_j are negated. The real parts x_j are not referenced.
- 2:** N — INTEGER *Input*
On entry: the number of data values, n .
Constraint: $N \geq 1$.
- 3:** IFAIL — INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1. For users not familiar with this parameter (described in Chapter P01) the recommended value is 0.
On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors detected by the routine:

IFAIL = 1
 N < 1.

7 Accuracy

Exact.

8 Further Comments

The time taken by the routine is negligible.

9 Example

This program reads in a sequence of real data values, calls C06EAF followed by C06GBF to compute their inverse discrete Fourier transform, and prints this after expanding it from Hermitian form into a full complex sequence.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```

*      C06GBF Example Program Text
*      Mark 14 Revised.  NAG Copyright 1989.
*      .. Parameters ..
      INTEGER          NMAX
      PARAMETER       (NMAX=20)
      INTEGER          NIN, NOUT
      PARAMETER       (NIN=5,NOUT=6)
*      .. Local Scalars ..
      INTEGER          IFAIL, J, N, N2, NJ
*      .. Local Arrays ..
      real            A(0:NMAX-1), B(0:NMAX-1), X(0:NMAX-1)
*      .. External Subroutines ..
      EXTERNAL         C06EAF, C06GBF
*      .. Intrinsic Functions ..
      INTRINSIC        MOD
*      .. Executable Statements ..
      WRITE (NOUT,*) 'C06GBF Example Program Results'
*      Skip heading in data file
      READ (NIN,*)
20    READ (NIN,*,END=100) N
      IF (N.GT.1 .AND. N.LT.NMAX) THEN
          DO 40 J = 0, N - 1
              READ (NIN,*) X(J)
40    CONTINUE
          IFAIL = 0
*
          CALL C06EAF(X,N,IFAIL)
          CALL C06GBF(X,N,IFAIL)
*
          WRITE (NOUT,*)
          WRITE (NOUT,*)
          +      'Components of inverse discrete Fourier transform'
          WRITE (NOUT,*)
          WRITE (NOUT,*) '          Real          Imag'
          WRITE (NOUT,*)
          A(0) = X(0)
          B(0) = 0.0e0
          N2 = (N-1)/2
          DO 60 J = 1, N2
              NJ = N - J
              A(J) = X(J)
              A(NJ) = X(J)
              B(J) = X(NJ)

```

```

        B(NJ) = -X(NJ)
60     CONTINUE
        IF (MOD(N,2).EQ.0) THEN
            A(N2+1) = X(N2+1)
            B(N2+1) = 0.0e0
        END IF
        DO 80 J = 0, N - 1
            WRITE (NOUT,99999) J, A(J), B(J)
80     CONTINUE
        GO TO 20
    ELSE
        WRITE (NOUT,*) 'Invalid value of N'
    END IF
100  STOP
*
99999 FORMAT (1X,I6,2F10.5)
END

```

9.2 Program Data

C06GBF Example Program Data

```

7
0.34907
0.54890
0.74776
0.94459
1.13850
1.32850
1.51370

```

9.3 Program Results

C06GBF Example Program Results

Components of inverse discrete Fourier transform

	Real	Imag
0	2.48361	0.00000
1	-0.26599	-0.53090
2	-0.25768	-0.20298
3	-0.25636	-0.05806
4	-0.25636	0.05806
5	-0.25768	0.20298
6	-0.26599	0.53090
